
Using the Food Patterns Equivalents (FPED) Database 
Instructions for FPED Examples 

 
Description of Examples 
 
The provided SAS code generates estimates for the mean amounts of food patterns cup equivalents 
of fruit consumed per individual in the United States using (a) What We Eat in America (WWEIA), 
NHANES August 2021 – August 2023, and (b) Food Patterns Equivalent Database (FPED) 2021 –
2023.   
 
FPED, a database produced and made publicly available by the USDA’s Food Surveys Research 
Group (FSRG), converts the foods and beverages in the Food and Nutrient Database for Dietary 
Studies (FNDDS) to 37 USDA Food Patterns components.  Further details on FPED and component 
definitions can be found in the 2021-2023 Food Patterns Equivalents Database Documentation file 
(1).  FNDDS provides the nutrient values for foods and beverages reported in WWEIA, the dietary 
intake component of NHANES (2, 3). 
 
FSRG uses WWEIA, NHANES and FPED to publish data tables reporting estimates for the mean 
amounts of food patterns equivalents consumed per individual in the United States for these 37 
food patterns components among various subpopulations. The FPED data tables, organized by 
NHANES cycle, are located at: https://www.ars.usda.gov/nea/bhnrc/fsrg/fpeddatatables (4).  
 
The examples presented here demonstrate how to use SAS to reproduce specific estimates found 
in these data tables.  SAS has analysis capabilities for handling survey data with a complex sample 
design.  See NHANES Tutorials - Sample Design Module for more details about the NHANES survey 
design and NHANES Tutorials - NHANES Dietary Analyses Module for more details specific to 
WWEIA, NHANES (5, 6). 
 

• Example 1: Replication of FPED Data Table 1a, Total Fruit, Stratified by Male/Female 
and Age 

 
The first example uses SAS to generate the estimated mean amounts of food patterns cup 
equivalents for total fruit consumed per individual, by male/female and age, in the United 
States, as seen in Table 1a of the FPED Data Tables. 

 
• Example 2: Replication of FPED Data Table 2a, All Fruit Components, Stratified by 

Race/Ethnicity and Age 
 

The second example uses SAS to generate the estimated mean amounts of food patterns 
cup equivalents for (a) total fruit, (b) citrus, melons, and berries, (c) other fruit, and (d) 
fruit juice consumed per individual, by race/ethnicity and age, in the United States, as 
seen in Table 2a of the FPED Data Tables.   
 

The examples are presented and discussed using SAS v. 9.4 (Cary, NC).  However, the steps could 
be adapted to other statistical software with complex survey capabilities. 

 

https://www.ars.usda.gov/nea/bhnrc/fsrg/fpeddatatables
https://wwwn.cdc.gov/nchs/nhanes/tutorials/SampleDesign.aspx
https://wwwn.cdc.gov/nchs/nhanes/tutorials/dietaryanalyses.aspx


2 
 

Example 1:  Replication of FPED Data Table 1A, Total Fruit, Stratified by 
Male/Female and Age 
 
We will be replicating the total fruit component of FPED Data Table 1a.  The provided SAS code will 
generate estimates of the means and standard errors by age group for all respondents, by age group 
for male respondents, and by age group for female respondents.  The code will generate a 
permanent SAS dataset and an Excel workbook containing the information from Table 1a circled 
below.  The organization of the SAS dataset and Excel workbook will be similar to that pictured, with 
columns for (a) male/female, (b) age group, (c) sample size, (d) mean for total fruit, and (e) standard 
error for total fruit and rows for each male/female and age group stratification. 

Step 1:  Download Data 
 
 
Navigate to the FPED Documentation and Databases webpage at: 
https://www.ars.usda.gov/northeast-area/beltsville-md-bhnrc/beltsville-human-nutrition-
research-center/food-surveys-research-group/docs/fped-doc-and-datasets/ . 
Select the FPED 2021-2023 tab and download the file FPED_DR1TOT_2123_sas.exe.   Also, 
download the Variable Descriptions File.  
 
Once FPED_DR1TOT_2123_sas.exe is downloaded, double-click on it to begin the data extraction 
process.  A window will open, prompting you to select a file location for permanent storage of the 
dataset FPED_DR1TOT_2123.sas7bdat.   
 

 
 
The examples use the location C:/ but you can designate a location of your choosing. Once you 
designate a location and click Extract, a copy of the dataset should be saved in that location. 
 
The dataset FPED_DR1TOT_2123.sas7bdat contains merged data from both WWEIA, NHANES and 
FPED.  It has one row per survey respondent for the WWEIA, NHANES day 1 dietary intake.  It 
includes data fields such as respondent age, race/ethnicity, ratio of family income to poverty, and 
day 1 dietary sample weights as well as the total day 1 consumption of 37 food patterns 
components.  For a complete list of variables and definitions, open and view the Variable 
Descriptions File that you also downloaded. 
 
 
 
Step 2:  Read in the Data and Recode Variables 
 
 
Read the dataset FPED_DR1TOT_2123.sas7bdat into SAS and prepare the data for further analysis.  
The code below defines a library named input as the location C:\.  The location should be changed 
if you indicated a location different than C:\ in Step 1.   
 
 

libname input "C:\"; 
 
 
Using a data step, create the working dataset data1 from input.FPED_DR1TOT_2123 (i.e., C:\ 
FPED_DR1TOT_2123).   
 
 

data data1; 
    set Input.FPED_DR1TOT_2123; 

https://www.ars.usda.gov/northeast-area/beltsville-md-bhnrc/beltsville-human-nutrition-research-center/food-surveys-research-group/docs/fped-doc-and-datasets/
https://www.ars.usda.gov/northeast-area/beltsville-md-bhnrc/beltsville-human-nutrition-research-center/food-surveys-research-group/docs/fped-doc-and-datasets/

Table 1a of the FPED Data Tables is generated for respondents at least two years of age with a 
reliable day 1 dietary intake record that did not consume breastmilk.  It is presented by age group 
and males/females.  The relevant variables in FPED_DR1TOT_2123.sas7bdat for identifying 
respondents for inclusion in the analysis, as described in the Variable Descriptions File, are:   
 
 

Variable  Description Code or Value 
DR1DRSTZ Dietary Recall Status, Day 1 1 = Reliable and met minimum criteria 

2 = Not reliable or not met minimum criteria 
4 = Reported consuming breastmilk 
5 = Not done 

DRABF Breast-fed infant (either day) 1 = Yes, infant breast fed on day 1 or day 2 
2 = No 

RIDAGEYR Age in years at screening 0 to 79 = Age in years, as entered 
80 = 80 years of age and over 

RIAGENDR Gender of the participant 1 = Male 
2 = Female 

 
 
Create the variable incohort to identify respondents for inclusion, where incohort = 1 if the 
respondent is included in the analysis for Table 1a (i.e., if RIDAGEYR ≥ 2 and DRABF ≠ 1 and 
DR1DRSTZ = 1) and incohort = 0 if the respondent is not included. 
 
 

    if (RIDAGEYR>=2) and (DRABF ne 1) and (DR1DRSTZ=1) then incohort = 1; 
    else incohort = 0; 

 
 

Capture the desired stratifications for age and male/female reflective of Table 1a by creating the 
new variables sex (male, female), agegroup1 (2-5, 6-11, 12-19, 20-29, 30-39, 40-49, 50-59, 60-69, 
70+ years), and agegroup2 (2-19, 20+ years).   
 
 

    sex = RIAGENDR;  
 
    if       2 <= RIDAGEYR <= 5     then agegroup1 = 1; 
    else if  6 <= RIDAGEYR <= 11    then agegroup1 = 2; 
    else if 12 <= RIDAGEYR <= 19    then agegroup1 = 3; 
    else if 20 <= RIDAGEYR <= 29    then agegroup1 = 4; 
    else if 30 <= RIDAGEYR <= 39    then agegroup1 = 5; 
    else if 40 <= RIDAGEYR <= 49    then agegroup1 = 6; 
    else if 50 <= RIDAGEYR <= 59    then agegroup1 = 7; 
    else if 60 <= RIDAGEYR <= 69    then agegroup1 = 8; 
    else if RIDAGEYR >= 70          then agegroup1 = 9; 
 
    if      ( 2 <= RIDAGEYR <= 19)   then agegroup2 = 1; 
    else if (RIDAGEYR >= 20)         then agegroup2 = 2; 
run; 

Step 3:  Generate the Weighted Mean Intakes of USDA Food Patterns Component  
 
 
Estimate the mean amount of food patterns cup equivalents for total fruit for the various age and 
male/female stratifications captured in Table 1a while accounting for the complex sample design 
used in the collection of WWEIA, NHANES.  For more detail on the complex sample design, see 
NHANES Tutorials - Sample Design Module (5).   
 
Start by estimating the mean amounts for all respondents in the entire cohort of interest (i.e., both 
male and female respondents ≥2 years of age).  The variable in FPED_DR1TOT_2123.sas7bdat 
which captures the day 1 total fruit food patterns component for each respondent is 
DR1T_F_TOTAL, defined as ‘Total intact fruits and fruit juices (cup eq)’. 
 
 

proc surveymeans nobs mean stderr data = data1; 
    strata SDMVSTRA;    
    cluster SDMVPSU;   
    weight WTDRD1; 
    domain incohort; 
    var DR1T_F_TOTAL; 
    ods output domain=est_all; 
run; 

 
 
 

https://wwwn.cdc.gov/nchs/nhanes/tutorials/SampleDesign.aspx

Defining each piece of the proc surveymeans code (7): 
 

Code Component Explanation 
proc surveymeans Indicates use of the SAS surveymeans procedure to compute 

means for data collected using a complex sample design. 

nobs mean stderr Indicates computation of the number of non-missing 
observations (nobs), mean (mean), and standard error (stderr). 

data = data1 Indicates computations should be performed on the temporary 
SAS dataset work.data1 (created in Step 3). 

strata SDMVSTRA Identifies SDMVSTRA (pseudo-stratum variable in NHANES) as 
the variable defining the first-level strata in the sample design.  

cluster SDMVPSU Identifies SDMVPSU (pseudo-PSU variable in NHANES) as the 
variable defining the clusters in the sample design. 

weight WTDRD1 Identifies WRDRD1 (day 1 dietary weight in WWEIA, NHANES) as 
the variable defining the sampling weight. 

domain incohort Indicates subpopulations for estimation.  Produces estimates by 
all levels of incohort, i.e., where incohort = 1 (desired 
subpopulation) and where incohort = 0 (will be ignored). 

var DR1T_F_TOTAL Identifies DR1T_F_TOTAL (FPED total fruit component) as the 
variable for which the number of non-missing observations, 
means, and standard errors will be calculated.   

ods output domain=est_all Identifies the output dataset (est_all) that will store the 
estimated nobs mean stderr for  DR1T_F_TOTAL  within the 
subpopulation(s) specified in the domain statement (incohort = 
1 and incohort = 0). 

 
 
Mean amounts of interest are shown in the Domains section of the surveymeans output. 
 

 
 
 
 
The output dataset created in the ods output statement, est_all, contains these same values.  
 

 
 

Estimates for incohort = 1.  
Since age group is not in 
the domain line, includes 
all respondents 2 years and 
older. Since sex is not in the 
domain line, includes 
males and females. 

For est_all, as well as all output datasets that will be created in future runs of the surveymeans 
procedure, we only want to retain estimates for the cohort of interest (i.e., where incohort = 1).  In 
addition, we want to create variables that will be the same across the output datasets that capture 
the age group (agegroup) and male/female (nsex)  stratifications seen in Table 1a.  The values to be 
assigned to the variables agegroup and nsex are shown below. 
 

Table 1a Structure 
Value of 

nsex 
Value of 

agegroup 
     Males: 1  
           2 – 5……………..  1 
         6  - 11 …………….  2 
       12  - 19 …………….  3 
       20  - 29 …………….  4 
       30  - 39 …………….  5 
       40  - 49 …………….  6 
       50  - 59 …………….  7 
       60  - 69 …………….  8 
       70  and over……….  9 
   
         2  - 19 …………….  10 
       20  and over……….  11 
          2  and over……….  12 
   
     Females: 2  
     …   
   
     Males and females: 3  
     …   

 
 
Create the dataset estimates_all to make these modifications to the output dataset est_all.   
 
 

data estimates_all; 
    set est_all; 
    if incohort=1; 
    agegroup = 12; 
    sex=3;  
run; 

 
 
Estimates_all looks similar to est_all, except that it only retains the rows where incohort = 1.  In 
addition, it has the variables nsex (assigned a value of 3 since the estimates are for males and 
females combined) and agegroup (assigned a value of 12 since the estimates are for respondents 2 
years and over). 

Continue estimating mean amounts for different subpopulations.  For example, the next 
surveymeans procedure produces estimates by agegroup2 (i.e., 2-19 and 20+ years). 
 
 

proc surveymeans nobs mean stderr data = data1; 
    strata SDMVSTRA;    
    cluster SDMVPSU;   
    weight WTDRD1; 
    domain incohort*agegroup2; 
    var DR1T_F_TOTAL; 
    ods output domain=est_age2; 
run; 

 
 
Notice the domain statement now includes the cross-product incohort*agegroup2, indicating 
that estimates will be produced for combinations of incohort and agegroup2 (i.e., all agegroup2 
categories where incohort=1 and all agegroup2 categories where incohort=0).  The Domains 
output contains the estimates of interest. 
 

 
 

 
 
 
 
 
 
The output dataset est_age2 contains these same results. 
 

  
Again, modify the output dataset to only include estimates for incohort = 1 and to add the variables 
nsex (assign a value of 3 to indicate the estimates are for males and females) and agegroup (assign 
values of 10 and 11 to reflect the 2-19 and 20+ age groups). 
 
 
 
 
 
 

Estimates for agegroup2 = 2 (20+ 
years) and incohort = 1.  Since sex 

does not appear in the domain line, 
this is for both males and females. 

Estimates for agegroup2 = 1 (2-19 
years) and incohort = 1.  Since sex 

does not appear in the domain line, 
this is for both males and females. 

Create the dataset estimates_ age2 to make these modifications to the output dataset est_ age2.   
 
data estimates_age2; 
    set est_age2; 
    if incohort=1; 

       if agegroup2=1 then agegroup = 10; 
     if agegroup2=2 then agegroup = 11; 

    nsex=3; 
run; 

 

 
 
 
Perform additional runs of the surveymeans procedure for the various age group and male/female 
stratifications in order to produce all the results needed to replicate the estimates from Table 1a.  
See the following page for additional code. 
 

/* All included respondents by sex. */ 
proc surveymeans nobs mean stderr data = data1; 
    strata SDMVSTRA;    
    cluster SDMVPSU;   
    weight WTDRD1; 
    domain incohort*sex; 
    var DR1T_F_TOTAL; 
    ods output domain=est_sex; 
run; 
data estimates_sex; 
    set est_sex; 
    if incohort=1; 
    agegroup = 12; 
    nsex=sex; 
run; 
 
/* All included respondents by sex and agegroup1. */ 
proc surveymeans nobs mean stderr data = data1; 
    strata SDMVSTRA;    
    cluster SDMVPSU;   
    weight WTDRD1; 
    domain incohort*sex*agegroup1; 
    var DR1T_F_TOTAL; 
    ods output domain=est_sexage1; 
run; 
data estimates_sexage1; 
    set est_sexage1; 
    if incohort=1; 
    agegroup = agegroup1; 
    nsex=sex; 
run; 

/* All included respondents by sex and agegroup2. */ 
proc surveymeans nobs mean stderr data = data1; 
    strata SDMVSTRA;    
    cluster SDMVPSU;   
    weight WTDRD1; 
    domain incohort*sex*agegroup2; 
    var DR1T_F_TOTAL; 
    ods output domain=est_sexage2; 
run; 
data estimates_sexage2; 
    set est_sexage2; 
    if incohort=1; 
    if agegroup2=1 then agegroup = 10; 
    if agegroup2=2 then agegroup = 11; 
    nsex=sex; 
run; 

 
 
 
 
 

Step 4:  Create Formats for the Demographic Groups of Interest 
 
 
The results from each surveymeans procedure will be aggregated into a common dataset in Step 5.  
In preparation, create formats for the variables agegroup (called agegroupf) and nsex (called sexf).   
The formats will be applied to in Step 5, essentially assigning labels to the variable values.   
 
 

proc format; 
    value agegroupf 
        1   = " 2 -  5    " 
        2   = " 6 - 11" 
        3   = "12 - 19" 
        4   = "20 - 29" 
        5   = "30 - 39" 
        6   = "40 - 49" 
        7   = "50 - 59" 
        8   = "60 - 69" 
        9   = "70 and over" 
        10  = "2-19" 
        11  = "20 and over" 
        12  = "2 and over"; 
 
    value sexf  
        1   = "Males  " 
        2   = "Females" 
        3   = "Males and females"; 
run; 

Step 5: Create Permanent SAS Dataset of Results Matching Table 1a 
 
 
The surveymeans procedures in Step 3 each produced a separate output dataset containing the 
estimates for the age group and male/female stratifications.  Create a new dataset, 
estimates_stacked, that combines all the output datasets together and drops unneeded variables. 
 
 

data estimates_stacked; 
    set estimates_all estimates_age2 estimates_sex estimates_sexage1  
        estimates_sexage2; 
    drop domainlabel incohort agegroup1 agegroup2 sex; 
run; 

 
 
The combined output dataset estimates_stacked contains the estimates for the mean amounts for 
all stratifications by male/female and age group, as desired.  However, the dataset structure does 
not yet match Table 1a.  In order to create a final dataset that matches the structure of Table 1a, the 
following modifications to estimates_stacked must be made: 
 

• Sort the rows by nsex and agegroup using proc sort.  
• Order the columns as they appear in Table 1a (nsex, agegroup, N, Mean, StdErr) using a 

retain statement. 
• Drop the variables VarName and VarLabel using a drop statement. 
• Apply the formats for nsex and agegroup, replacing numeric values with format labels, 

using a format statement. 
• Apply formats to mean and stderr to present the variables with 2 and 3 decimals, 

respectively. 
 
Use a data step to make these changes, saving a permanent SAS dataset FPED_Ex1 in the library 
input (previously defined as the location C:\).  
 
 

proc sort data=estimates_stacked; 
    by nsex agegroup; 
run; 
data input.FPED_Ex1; 
    retain nsex agegroup N mean stderr; 
    set estimates_stacked; 
 
    *Remove variables no longer needed; 
    drop VarName; 
 
    *Apply formats to nsex and agegroup variables; 
    format nsex sexf. agegroup agegroupf.; 
 
    *Present mean/std error values to two/three decimal places; 
    format mean 8.2 stderr 8.3; 
run; 

Navigate to the location specified by your input library and verify that the final, permanent SAS 
dataset FPED_Ex1 appears.  This dataset contains the estimates of the mean amounts of food 
patterns cup equivalents for total fruit consumed per individual, by male/female and age, in the 
United States, as seen in Table 1a of the FPED Data Tables.   

Step 6:  Export the Data into Microsoft Excel (optional)  
 
 
If desired, export the data into Excel.  SAS has an export procedure; however, no variable formatting 
is retained.  Because we wish to retain the formats agegroupf and sexf added to the variables 
agegroup and nsex, apply ODS excel to export the data (8). 
 
 

ods excel file="C:\FPED_Ex1.xlsx"; 
    ods excel options(sheet_name="Table 1A, Total Fruit"); 
    proc print data=input.FPED_Ex1 noobs; 
        title "Mean Cup Equivalents of Total Fruit Consumed Per  
        Individual, by Male/Female and Age, in the United States,  
        August 2021-August 2023"; 
    run; 
ods excel close; 

 
 

Defining each piece of the ods excel code (8): 
 

Code Component Explanation 
ods excel file="C:\ 
FPED_Ex1.xlsx " 

Opens ODS excel, such that the output of all subsequent code will 
be sent to Excel, specifically to the file location specified, 
C:\FPED_Ex1.xlsx.  The file location and name can be changed by 
the user if desired. 

ods excel 
options(sheet_name="Table 
1A, Total Fruit")  

Generates a name ‘Table 1a, Total Fruit’ for the excel worksheet 
within the file  C:\FPED_Ex1.xlsx where the output will be sent.  

proc print 
data=input.FPED1A_fruit 
noobs; 

Prints the final permanent SAS dataset created in Step 5.  Thus, 
basically generates a copy of the dataset for export.  Noobs 
removes the default observations numbers from each row. 

title "Mean Cup ... -August 
2023"; 

Adds a title to the SAS output and to the generated Excel file.  This 
title is viewable in Excel in print mode. 

ods excel close Closes ODS excel, ending the export of output to Excel. 
 
 
After running this code, an excel file of the data will open on your screen.  In addition, the Excel file 
FPED_Ex1 will appear in the location specified in the ods excel file location. 

Example 2:  Replication of FPED Data Table 2A, Total Fruit, 
Citrus/Melons/Berries, Other Fruit, and Fruit Juice, Stratified by 
Race/Ethnicity and Age 
 
We will explore how Example 1 would change if we wanted to estimate mean amounts for several 
FPED components at once and across different subgroups of interest.  We will be replicating 
elements of FPED Data Table 2a by estimating mean amounts of all four fruit components: (a) total 
fruit, (b) citrus, melons, berries, (c) other fruit, and (d) fruit juice and stratifying the results by 
race/ethnicity and age rather than male/female and age.  The provided SAS code will generate 
estimates of the means and standard errors by age group within race/ethnicity groups.  The code 
will generate a permanent SAS dataset and an Excel workbook containing the information from 
Table 2a circled below.  The organization of the SAS dataset and Excel workbook will be similar to 
that pictured, with columns for (a) race/ethnicity, (b) age group, (c) sample size, ((d) mean for total 
fruit, (e) standard error for total fruit, (f) mean for citrus, melons, berries, (g) standard error for 
citrus, melons, berries, (h) mean for other fruit, (i) standard error for other fruit, (j) mean for fruit 
juice, and (k) standard error for fruit juice and rows for each race/ethnicity and age group 
stratification. Relative to FPED Data Table 1a, some age groups are collapsed due to small sample 
sizes. 

Step 1:  Download Data 
 
 
See Step 1 from Example 1. 
 
 
 
 
Step 2:  Read in the Data and Recode Variables 
 
 
Create a library and the working dataset data2 from input.FPED_DR1TOT_2123, as described in 
Example 1.   
 
 

libname input "C:\"; 
data data2; 
    set Input.FPED_DR1TOT_2123; 

 
 
Table 2a of the FPED Data Tables is presented by race/ethnicity and age group.  The variable in 
FPED_DR1TOT_2123.sas7bdat capturing race/ethnicity, as described in the Variable Descriptions 
File, is:   
 

Variable  Description Code or Value 
RIDRETH3 Race/Hispanic origin with 

Non-Hispanic Asian 
1 = Mexican American 
2 = Other Hispanic 
3 = Non-Hispanic White 
4 = Non-Hispanic Black 
6 = Non-Hispanic Asian 
7 = Other Race – Including Multi-Racial 

 
 
As done in Example 1, create the variable incohort to identify respondents for inclusion in the 
analysis and define the variables capturing the age and race/ethnicity stratifications in Table 2a.  
Also create data labels for the variables capturing all the fruit components. 
 
 

    if (RIDAGEYR>=2) and (DRABF ne 1) and (DR1DRSTZ=1) then incohort=1; 
    else incohort=0; 
 
    if      ( 2 <= RIDAGEYR <= 19)   then agegroup2 = 1; 
    else if (RIDAGEYR >= 20)         then agegroup2 = 2; 
 
    if      RIDRETH3 = 3            then race_eth = 1;    
    else if RIDRETH3 = 4            then race_eth = 2;    
    else if RIDRETH3 = 6            then race_eth = 3;    
    else if RIDRETH3 in(1 2)        then race_eth = 4;    
    else                                 race_eth = 5;    

    label DR1T_F_TOTAL          = "Total_Fruit"; 

         label DR1T_F_CITMLB         = "Citrus_Melons_Berries"; 
         label DR1T_F_OTHER          = "Other_Fruit"; 
         label DR1T_F_JUICE          = "Fruit_Juice"; 

run; 
 
 
 
 

Step 3:  Generate the Weighted Mean Intakes of USDA Food Patterns Components  
 
 
As done in Example 1, estimate the mean amount of food patterns cup equivalents for total fruit, 
citrus/melons/berries, other fruit, and fruit juice for the various age and race/ethnicity 
stratifications captured in Table 2a while accounting for the complex sample design used in the 
collection of WWEIA, NHANES.   
 
The relevant original variables in FPED_DR1TOT_2123.sas7bdat capturing  the day 1 component 
totals for each respondent, as described in the Variable Descriptions File, are: 
 

Variable  Description (Day 1) Value 
DR1T_F_TOTAL Total intact fruits and fruit juices (cup eq.) Numeric  
DR1T_F_CITMLB Intact fruits of citrus, melons, and berries (cup eq.) Numeric  
DR1T_F_OTHER Intact fruits; not citrus, melons, and berries (cup eq.) Numeric  
DR1T_F_JUICE Fruit juices, citrus and non-citrus (cup eq.) Numeric  

 
 
Start by estimating the mean amount for all respondents in the cohort of interest (i.e., all 
respondents ≥2 years of age) by race/ethnicity.  Modify the var statement in the surveymeans 
procedures to include the four fruit component variables.  Also, change the variables listed in the 
domain statement, using race_eth rather than sex. 
 
 

proc surveymeans nobs mean stderr data = data2; 
    strata SDMVSTRA;    
    cluster SDMVPSU;   
    weight WTDRD1; 
    domain incohort*race_eth; 
    var DR1T_F_TOTAL DR1T_F_CITMLB DR1T_F_OTHER DR1T_F_JUICE; 
    ods output domain=est_race; 
run; 

The output dataset created in the ods output statement, est_race, contains the mean amounts for 
each fruit component by race/ethnicity for all respondents at least 2 years of age:  
 

  
  

 
 Estimates for race_eth = 1 (Non-Hispanic White) and incohort = 1 for each fruit component.  

Since agegroup2 does not appear in the domain line, this is for respondents 2 years and over. 

Estimates for race_eth = 2 (Non-Hispanic Black)  and incohort = 1 for each fruit component.  
Since agegroup2 does not appear in the domain line, this is for respondents 2 years and over. 

Estimates for race_eth = 3 (Non-Hispanic Asian) and incohort = 1 for each fruit component.  
Since agegroup2 does not appear in the domain line, this is for respondents 2 years and over. 

Estimates for race_eth = 4 (Hispanic)  and incohort = 1 for each fruit component.  Since 
agegroup2 does not appear in the domain line, this is for respondents 2 years and over. 



19 
 

As in Example 1, a common variable named agegroup will be created to take on the age group 
stratifications in Table 2a for all the results datasets (and race_eth values will be retained, as they 
already match the stratifications in Table 2a).  The variable agegroup will take on the following 
values: 
 

Table 2a Structure 
Value of 
race_eth 

Value of 
agegroup 

     Non-Hispanic White: 1  
         2  - 19 …………….  1 
       20  and over……….  2 
          2  and over……….  3 
   
     Non-Hispanic Black: 2  
         2  - 19 …………….  1 
       20  and over……….  2 
          2  and over……….  3 
   
     Non-Hispanic Asian: 3  
     …   
   
     Non-Hispanic Asian: 4  
     …   

 
Create the dataset estimates_race, as done in Example 1, retaining those respondents in the cohort 
of interest (incohort = 1), excluding respondents in the ‘other’ race/ethnicity category, and assigning 
a value of 3 to agegroup (since agegroup2 was not included in the domain statement).  Continue 
estimating the mean amounts by race/ethnicity and by age group. 
 

data estimates_race; 
    set est_race; 
    if incohort=1 and race_eth^=5; 
    agegroup=3; 
run; 
 
/* All included respondents by race/ethnicity and agegroup2. */ 
proc surveymeans nobs mean stderr data = data2; 
    strata SDMVSTRA;    
    cluster SDMVPSU;   
    weight WTDRD1; 
    domain incohort*race_eth*agegroup2; 
    var DR1T_F_TOTAL DR1T_F_CITMLB DR1T_F_OTHER DR1T_F_JUICE; 
    ods output domain=est_raceage2; 
run; 
data estimates_raceage2; 
    set est_raceage2; 
    if incohort=1 and race_eth^=5; 
    agegroup = agegroup2; 
run; 

Step 4:  Create Formats for the Demographic Groups of Interest 
 
 
As done in Example 1, create the formats for agegroup (called agegroupf) and race_eth (called 
race_ethf). 
 

proc format; 
    value agegroupf 
        1  = "2-19       " 
        2  = "20 and over" 
        3  = "2 and over"; 
 
    value race_ethf  
        1   = "Non-Hispanic White" 
        2   = "Non-Hispanic Black" 
        3   = "Non-Hispanic Asian" 
        4   = "Hispanic"; 
run; 

 
 
 
Step 5: Create Permanent SAS Dataset of Results Matching Table 2a 
 
 
The surveymeans procedures in Step 3 each produced a separate output dataset of results for the 
various stratifications of age group and race/ethnicity.  Create a new dataset, estimates_stacked, 
that combines all the output datasets together and drops unneeded variables. 
 
 

data estimates_stacked; 
    set estimates_race estimates_raceage2; 
    drop domainlabel incohort agegroup2; 
run; 

 
 
The dataset estimates_stacked contains the estimates for the mean amount of each fruit 
component for all combinations of race/ethnicity and age group.  The first nine lines are shown 
below.  However, as with Example 1, the dataset structure does not yet match Table 2a.  Additional 
modifications are needed. 

In order to create a final dataset that matches the structure of Table 2a, the following changes to 
estimates_stacked must be made: 
 

• Sort the rows by race_eth and agegroup.  
• Transpose the data so that there is a column for Mean and StdErr for each fruit component. 
• Order the columns as they appear in Table 2a (race_eth, agegroup, N, Mean for Total Fruit, 

StdErr for Total Fruit, Mean for Citrus/Berries/Melons, StdErr for Citrus/Berries/Melons, 
Mean for Other Fruit, StdErr for Other Fruit, Mean for Fruit Juice, StdErr for Fruit Juice). 

• Drop the variables VarName and VarLabel. 
• Apply the formats for race_eth and agegroup, replacing numeric values with category 

labels. 
 
Use the sort procedure to sort the rows by race_eth and agegroup and then use the transpose 
procedure to create columns of the number of observations, N, for each component. 
 

proc sort data=estimates_stacked; 
    by race_eth agegroup VarLabel; 
run; 
 
proc transpose data=estimates_stacked out=FPED_N prefix=N_; 
    by race_eth agegroup;  
    var N; 
    id VarLabel; 
run; 

 
 
Defining each piece of the transpose code (9):   
 

Code Component Explanation 
proc sort 
data=estimates_stacked 

Sorts the temporary SAS dataset work.estimates_stacked.  
Sorting is necessary in order to use a by statement when 
transposing the data. 

by race_eth agegroup VarLabel Sorts estimates_stacked by the variables race_eth, agegroup, 
and VarLabel (the names of the four fruit components).   

proc transpose 
data=estimates_stacked 

Transposes elements of the SAS dataset estimates_stacked, 
creating additional columns from the rows. 

out=FPED_N prefix=N_ Stores the new transposed dataset in a new dataset FPED_N.  
Assigns the prefix ‘N_’ to all the transposed variables. 

by race_eth agegroup Forms groups by race_eth and agegroup when transposing. 

var N Transposes the variable N, the number of observations.  The 
transpose procedure will be repeated for the means and 
standard errors. 

id VarLabel Indicates the variable VarLabel should be used for naming the 
variables in the output dataset. 

 



22 
 

The resulting dataset FPED_N retains rows for the various combinations of race/ethnicity and age 
group but contains the number of observations, N, for each of the four fruit components in a 
separate column, with the variable appearing as N_fruit_component_name. 
 

 
 
 
Use additional transpose statements to change the structure of the datasets containing the means 
and standard deviations.   
 
 

/* Repeat for the means. */ 
proc transpose data=estimates_stacked out=FPED_mean prefix=Mean_; 
    by race_eth agegroup; 
    var mean; 
    id VarLabel; 
run; 
 
/* Repeat for the standard errors. */ 
proc transpose data=estimates_stacked out=FPED_stderr prefix=stderr_; 
    by race_eth agegroup; 
    var stderr; 
    id VarLabel; 
run; 

Merge the three transposed datasets (FPED_N, FPED_mean, and FPED_stderr) together by 
race_eth and agegroup.  Use a data step to retain the needed variables in the correct order and 
format race_eth and agegroup.  Then save a permanent SAS dataset FPED_Ex2 in the library input 
(previously defined as the location C:\).  
 
 

data FPED; 
    merge FPED_N FPED_mean FPED_stderr; 
    by race_eth agegroup; 
    drop _NAME_ _LABEL_; 
run; 
data input.FPED_Ex2; 
    retain race_eth agegroup N_total_fruit mean_total_fruit  
    stderr_total_fruit 
    mean_citrus_melons_berries stderr_citrus_melons_berries 
    mean_other_fruit stderr_other_fruit 
    mean_fruit_juice stderr_fruit_juice; 
    set FPED; 

 
    *Apply a common sample size name (N) to the sample size variable; 
    rename N_total_fruit=N; 
 
    *Remove variables no longer needed; 
    drop N_citrus_melons_berries N_fruit_juice N_Other_fruit; 
 
    *Apply formats to nsex and agegroup variables; 
    format race_eth race_ethf. agegroup agegroupf.; 
 
    *Present mean values to two decimal places and standard errors to  
    three decimal places; 
    format mean_total_fruit 8.2 mean_citrus_melons_berries 8.2 
    mean_other_fruit 8.2 mean_fruit_juice 8.2  
    stderr_total_fruit 8.3 stderr_citrus_melons_berries 8.3 
    stderr_other_fruit 8.3 stderr_fruit_juice 8.3; 
run; 

 
 
Navigate to the location specified by your input library and verify that the final, permanent SAS 
dataset FPED_Ex2 appears.  The final, permanent SAS dataset input.FPED_Ex2 contains the 
estimates of the mean amounts of food patterns cup equivalents for all fruit components 
consumed per individual, by race/ethnicity and age, in the United States, as seen in Table 2a of the 
FPED Data Tables.   

Step 6:  Export the Data into Microsoft Excel (optional)  
 
 
Export the data to Excel using a variation of the code from Example 1,specifying new dataset and 
file names. 
 
 

ods excel file="C:\FPED_Ex2.xlsx"; 
    ods excel options(sheet_name="Table 2A, Total Fruit"); 

     proc print data=input.FPED_Ex2 noobs; 
  title "Mean Cup Equivalents of Total Fruit, Citrus/Melons/Berries,  
  Other Fruit, and Fruit Juice Consumed per Individual, 
  by Race/Ethnicity and Age, in the United States,  
  August 2021-August 2023"; 
     run; 

ods excel close; 
 
 
As previously explained, after running this code, an excel file of the data will open on your screen.  
In addition, the Excel file FPED_Ex2 will appear in the location specified in the ods excel file 
location. 

